

# Equilibrium muscle cross-bridge behavior

## Theoretical considerations II

### Model describing the behavior of strongly-binding cross-bridges when both heads of myosin bind to the actin filament

Mark Schoenberg

Laboratory of Physical Biology, National Institute of Arthritis and Musculoskeletal and Skin Diseases, National Institutes of Health, Bethesda, Maryland 20892 USA

**ABSTRACT** A model has been developed for characterizing the interaction between strongly-binding myosin cross-bridges and actin in muscle fibers under equilibrium conditions where both heads of the myosin cross-bridge bind to actin. The model, that of Anderson and Schoenberg (1987, *Biophys. J.* 52:1077–1082) is quite similar to that of Schoenberg (1985, *Biophys. J.* 48:467–475), except that explicit account is taken of the fact that each crossbridge has two heads which can bind to actin. The key assumption that allows this model to explain a large body of data unexplained by the Schoenberg (1985) model is that the two crossbridge heads are not totally independent of one another after attachment. After the first head attaches, the second head is then free to attach only to an actin site distal to the first head. This means that when the more distally attached head subsequently detaches and reattaches (as the heads continually do), it will not reattach in a position of lesser strain and reduce the force it supports, but instead will remain attached in its strained position until the proximally attached head also detaches. This model gives an explanation for two important and otherwise unexplained observations made previously: it explains why at ionic strengths in the range of 50–120 mM, (a) the rate constant of force decay after a small stretch is a sigmoidal function of nucleotide analogue concentration, and (b) why in the presence of analogues or in rigor the rate constant of force decay after a small stretch is significantly slower than the rate constant for myosin subfragment-1 detachment from actin in solution.

## INTRODUCTION

In the mid-1970's T. L. Hill (Hill, 1974; 1975) postulated that the proteins that interact in intact skeletal muscle fibers to produce muscle shortening and force must follow the same sort of thermodynamic principles and relationships that govern protein interactions in solution. He also discussed a number of models that might be useful for describing the interaction between actin and myosin in muscle fibers. In 1985, I used one of these models, the model of a single-headed cross-bridge looking at an infinite array of equivalent actin sites, to examine how muscle fibers might respond to various experimental maneuvers. Despite the simplicity of the model, it was very successful in describing much of the behavior of real muscle fibers (Schoenberg and Eisenberg, 1985; Schoenberg, 1988a,b; 1989). Although the model quantitatively explained the behavior of the weakly-binding myosin-ATP crossbridge (Schoenberg, 1988a, b), it had two deficiencies with regard to predicting the behavior of strongly-binding cross-bridges. For one, it failed to explain why the apparent detachment rate constants of the AMPPNP, PP<sub>i</sub>, ADP, or rigor cross-bridges are significantly slower than the corresponding subfragment-1 detachment rate constants in solution. Secondly, it did not predict the finding of Anderson and Schoenberg (1987) that the rate constant of force decay after a small stretch in the presence of MgPP<sub>i</sub> at ionic strength 110 mM is a sigmoidal rather than a Michaelis function of [MgPP<sub>i</sub>].

The current work extends the 1985 modeling of equilibrium cross-bridge behavior from the simple case where the crossbridges are considered to have only a single head or independent heads, to the case where the cross-bridge is considered to have two heads capable of binding to actin. Significantly, this model, which is meant to describe the behavior of so-called strongly-binding cross-bridges, postulates an interaction between the heads. This is crucial to the success of the model in explaining the experimental data because a previous model of double-headed cross-bridges which did not postulate an interaction between the heads (Tozeren and Schoenberg, 1986) gave behavior very similar to that of the simple 1985 single-headed cross-bridge model. The current model differs from these earlier models in that it correctly predicts the sigmoidal shape of the relationship between the rate constant of force decay after a small stretch and [MgPP<sub>i</sub>] (Anderson and Schoenberg, 1987), and it also explains why the rate constant of force decay after a small stretch is usually less than the corresponding detachment rate constant of myosin subfragment-1 from actin in solution.

## METHODS

The technique for computing the behavior of equilibrium cross-bridge models has been described previously (Schoenberg, 1985; Tozeren and Schoenberg, 1986).

The current model, like previous models (Schoenberg, 1985; Tozeren and Schoenberg, 1986), makes the simplifying assumption that all the actin sites are identical and equivalent. This is tantamount to assuming that there is enough azimuthal flexibility in the binding of the cross-bridge head that the helicity present in the actin filament does not present a severe impediment to cross-bridge binding. This is not as restrictive an assumption as it seems, because, for the models considered, the energy of binding and the cross-bridge stiffness are such that the cross-bridges generally bind within one actin site of their least-strained position.

At ionic strengths where heads without nucleotide are tightly bound to actin the simplest scheme for detachment of the double-headed cross-bridge is as follows:



Turning now to the attachment rate constants: if we define the attachment rate constant of an isolated head as  $k_t$ , again from the law of mass action, we can assume that the attachment rate constants of steps 3, 4, and 6 are  $k_t$ ,  $k_t$ , and  $2k_t$ , respectively. Considering first the case where the cross-bridge is unstrained: if the detachment rate constant of an isolated cross-bridge head is  $k_b$  and the strength of binding of the head is  $K_b$ , the attachment rate constant for zero strain will be  $k_t = K_b k_b$ . The value of  $k_t$  when strain is not zero can be determined from Hill's formalism (Hill, 1974) provided assumptions are made about the stiffness of each attached crossbridge head. Two limiting assumptions about cross-bridge head stiffness are made. With the first assumption, corresponding to model I of Tozeren (1987), it is assumed that

binding of the second cross-bridge head does not significantly increase overall cross-bridge stiffness. The second limiting assumption, corresponding to Model II of Tozeren (1987), is that binding of the second head doubles overall cross-bridge stiffness. Either of these assumptions makes the attachment rate constant a function of the strain coordinate, as shown in Appendix I.

With these assumptions about the cross-bridge attachment and detachment rate constants, the model is fully defined once assumptions about the rate constants of nucleotide binding also are made. Defining the attachment and detachment rate constants of nucleotide to an attached head as  $k_+$  and  $k_-$ , the model assumes that the forward and back rate constants of steps 1, 2, and 5 are  $2k_+[N]$ ,  $k_+[N]$ ,  $k_+[N]$  and  $k_-$ ,  $2k_-$ , and  $k_-$ , respectively. These rate constants are assumed independent of cross-bridge strain. All that remains is to define the magnitudes of  $k_f$ ,  $k_b$ ,  $k_+$ , and  $k_-$ .

The kinetic scheme, Scheme I, is the simplest conceivable kinetic scheme since both nucleotide and cross-bridge binding are assumed one step processes. In reality, we know from experimental work of Trybus and Taylor (1982) and Coates et al. (1985) that both these events are actually two-step processes. First, there is rapid binding of nucleotide, then a rate-limiting conformational change, and finally, a very rapid detachment of the cross-bridge head.

To make Scheme I compatible with this experimental picture while avoiding the necessity of adding an extra step to represent the very rapid detachment of the cross-bridge head, we make  $k_+$  and  $k_-$  very large compared to  $k_b$ , and we lump the relatively slow conformational change and very rapid detachment into one slow detachment step of rate constant  $k_b$ . This makes a relatively simple computational model compatible with the known experimental data. The specific values of all variables and the equations used in the computations are given in Appendix I.

The set of linear differential equations generated by the above assumptions is easily solved using Gear's method (Gear, 1971). The computations were done on a model 3/260 workstation (Sun Microsystems, Sun Valley, CA) having a 68881 floating point microprocessor chip. The most common calculation, computing the time course of the decay in force after a small stretch, required, depending upon the parameters, between 30 and 100s of machine time.

## RESULTS

In many respects, the behavior of the current double-headed model is similar to that of the Schoenberg (1985)

single-headed model. The results presented here concentrate mainly upon those aspects of model behavior that are different from the behavior of the simpler single-headed cross-bridge model. The two major areas where the behavior of the current model is significantly different from the simpler single-headed model are the nucleotide concentration dependence of the rate constant of force decay after a small stretch, and the dependence of this rate constant upon the strength of binding of the cross-bridge heads.

We first examine the behavior of the double-headed model when head binding strength is weak. For this condition, most of the time only a single crossbridge head is attached so that if our computations are correct the current model should behave very much like the simpler single-headed 1985 model. We then examine the behavior of the model when head binding is strong. Here, the behavior of the double-headed model is significantly different from that of the simpler single-headed model; only it is totally compatible with the existing data. Lastly, we examine in detail how the rate constant for force decay after a small stretch varies with strength of binding of the cross-bridge heads.

### Force decay after stretch with weak head binding

Fig. 1 shows two force decay curves calculated for the double-headed model when head binding is weak ( $K_b = 10^{-2}$ ). The solid curve shows the decay when nucleotide concentration is equal to the dissociation constant for nucleotide binding and the dashed curve




FIGURE 1 The time course of force decay after a small stretch with weak head binding ( $K_b = 10^{-2}$ ). The force immediately after stretch (time = 0) is normalized to 1.0. The solid curve shows the calculations for the case where the nucleotide concentration is equal to the dissociation constant for nucleotide binding and the dashed curve shows the computed result for nucleotide concentration one half that amount.

shows the decay when nucleotide concentration is one half that amount. Each decay is described by a single exponential, a consequence of the fact that the cross-bridge head detachment rate constant is assumed independent of cross-bridge strain. By calculating several curves such as those in Fig. 1, it is possible to construct the nucleotide concentration dependence of the rate constant of force decay for weak head binding. This is shown in Fig. 2.

Although Michaelis-Menton kinetics was derived for the steady state and the term Michaelis-Menton perhaps should not be applied to kinetic results such as those in Fig. 1, nonetheless, it was derived in Anderson and Schoenberg (1987) that when the cross-bridge heads act totally independently, the dependence of the rate constant of force decay on nucleotide or nucleotide analogue concentration should have a shape described by the Michaelis-Menton equation. Thus, the solid curve in Fig. 2 shows a Michaelis function with a  $V_{\max}$  of  $r_0$  and a Michaelis constant of  $K_d = k_-/k_+$ . It is seen that when head binding is weak, the behavior of the double-headed model is virtually identical to that of the single-headed model, the nucleotide concentration dependence of the rate constant of force decay following a Michaelis-Menten relationship with  $V_{\max} = r_0$ .




FIGURE 2 Calculated concentration dependence of the rate constant of force decay for the double-headed cross-bridge model with weak cross-bridge head binding ( $K_b = 10^{-2}$ ).  $[N]$  is nucleotide concentration and  $K_d = k_-/k_+$  is the dissociation constant for  $N$  binding to an attached cross-bridge head. The symbols are the calculated rate constants for force decay ( $r$ ), normalized by the rate constant for detachment of a single isolated head ( $r_0$ ). The solid curve shows the Michaelis-Menten equation for  $V_{\max} = r_0$  and Michaelis constant  $= K_d$ . Note that the calculated results are described by the solid curve quite well.

## Force decay after stretch with strong head binding

Fig. 3 shows two computed force decays with strong head binding ( $K_b = 10^2$ ). The solid curve shows the decay when nucleotide concentration is equal to the dissociation constant for nucleotide binding and the dashed curve shows the decay when nucleotide concentration is one-half that amount. In contrast to the case where head binding is weak (Fig. 1), the computed results in Fig. 3 are quite different from those given by the simple single-headed model. For the solid curve in Fig. 3, the half-time for force decay is  $k_b t = 40.8$ , much longer than  $k_b t = 1.38$  expected from the single-headed model for  $[N]/K_d = 1$ . The computations for Fig. 3 were done with the assumption that binding of the second head contributes as much to cross-bridge stiffness as binding of the first head. Similar, but not identical results are obtained with the assumption that binding of the second head contributes little to cross-bridge stiffness (see below).

By doing additional computations such as those in Fig. 3, it is possible to construct the nucleotide dependence of the rate constant of force decay with strong head binding. This is shown in Fig. 4 where one sees a relationship very different from that of previous models. In previous models, as stated earlier, the relationship




FIGURE 3 The time course of force decay after a small stretch with strong head binding ( $K_b = 10^2$ ). The force immediately after stretch (time = 0) has been normalized to 1.0. The solid curve shows the calculations for the case where the nucleotide concentration is equal to the dissociation constant for nucleotide binding and the dashed curve shows the computed result for nucleotide concentration one half that amount. Note that the force decay curves are each described by a single exponential but that the half times for force decay,  $k_b t = 40.8$  and 90.6 are considerably longer than those expected from the single-headed model (Schoenberg, 1985),  $k_b t \approx 1.4$  and 2.1.




FIGURE 4 Calculated concentration dependence of the rate constant of force decay for the double-headed cross-bridge model with strong cross-bridge head binding ( $K_B = 10^3$ ). The symbols are the calculated rate constants for force decay ( $r$ ), normalized by the rate constant for detachment of a single isolated head ( $r_0$ ).  $[N]$  is nucleotide concentration and  $K_d = k_-/k_+$  is the dissociation constant for  $N$  binding to an attached cross-bridge head. The solid curve is the best fit of Eq. 1 with  $n = 2$  to the calculated theoretical rate constants of decay. The dashed curve is the best fit of a Michaelis-Menten expression (Eq. 1 with  $n = 1$ ) to the theoretical rate constants.

between the rate constant of force decay and nucleotide concentration always was mathematically described by the Michaelis-Menten equation. The relationship for the current model with strong head binding is sigmoidal, fitted not by the Michaelis equation, but instead by Eq. 3 of Anderson and Schoenberg (1987) with  $n = 2$ . The best fit of a Michaelis-Menten equation to the relationship, clearly a very poor fit, is shown by the dashed curve in Fig. 4. The excellent fit of the equation of Anderson and Schoenberg with  $n = 2$  is shown by the solid curve.

Eq. 3 of Anderson and Schoenberg (1987) states that if  $n$  heads must bind nucleotide before any head can relax the tension it supports, then the relationship between the rate constant of tension decay,  $r$ , and nucleotide concentration,  $[N]$ , will be

$$r/r_{\max} = \frac{[N]^n}{(K_d + [N])^n}, \quad (1)$$

where  $K_d$  is the dissociation constant for nucleotide binding. When  $n = 1$ , the above equation reduces to the Michaelis-Menten equation.

It should be noted that the computed results shown in Fig. 4, whereas quite different from those of the simpler single-headed model, agree extremely well with available experimental data. Of particular note are that with

the double-headed model the rate constants for force decay are significantly less than the rate constant for subfragment-1 detachment from actin in solution (Clarke and Tregear, 1982) and the nucleotide concentration dependence is sigmoidal (Anderson and Schoenberg, 1987).

### Rate constant of force decay at infinite nucleotide concentration as a function of head binding strength

We have already seen that with weak head binding and high nucleotide concentration the rate constant of force decay equals the rate constant for detachment of an isolated head, and with  $K_B = 10^2$  the rate constant of force decay is significantly slower than this. By doing additional calculations at high nucleotide concentration over a range of head binding strengths, it is possible to calculate the dependence of the rate constant for force decay on head binding constant. This is shown in Fig. 5.




FIGURE 5 The theoretical dependence of the rate constant of force decay after stretch in the fiber as a function of the cross-bridge head binding constant,  $K_B$ . The ordinate is the logarithm of the calculated rate constant of force decay after stretch,  $r$ , normalized by the detachment rate constant of a single isolated head,  $r_0$ .  $r_0$  is postulated to be equal to the detachment rate constant of myosin subfragment-1 from actin in solution. Because the calculations were done at exceedingly high nucleotide concentration ( $[N]/K_d = 10^3$ ),  $r$  is equal to  $r_{\max}$  of Eq. 1. The solid curve gives the calculated relationship assuming binding of second head contributes as much to overall cross-bridge stiffness as binding of first head. The dashed curve gives the relationship assuming binding of the second head does not increase overall cross-bridge stiffness. The important points are that with the double-headed model the calculated rate constant of force decay is very dependent upon head binding strength and only with very weak cross-bridge binding does the theoretical rate constant for force decay after stretch approach the rate constant for subfragment-1 detachment from actin.

Tozeren (1987) previously pointed out that a factor influencing the behavior of the double-headed cross-bridge is whether attachment of the second head contributes the same amount to cross-bridge stiffness as attachment of the first head. Thus, in Fig. 5, the solid theoretical curve is calculated using the assumption that both cross-bridge heads contribute the same amount of stiffness when they bind and the dashed theoretical curve shows the computed response assuming that binding of the second head does not increase overall cross-bridge stiffness.

It is seen from Fig. 5 that for both cases considered, the rate constant of force decay after stretch ( $r$ ) is very dependent upon the cross-bridge head binding constant ( $K_b$ ). Assuming that a single isolated myosin head detaches with the same rate constant as myosin subfragment-1 detaches from actin in solution, the rate constant of force decay is equal to the rate constant of subfragment-1 detachment from actin in solution when head binding is very weak (i.e., when the cross-bridges are attached by only a single head), but is very much slower when head binding is strong. As suggested by Tozeren, there are slight quantitative differences in behavior depending upon the contribution of the second head to overall stiffness, but essentially, the effect of head binding strength is similar in both cases.

## DISCUSSION

Whereas the very simple model of Schoenberg (1985) explains virtually all of the behavior of cross-bridges with ATP at the nucleotide binding site, it has two notable shortcomings in describing the behavior of more strongly-binding cross-bridges. It does not explain why in rigor and with ATP analogues at the nucleotide binding site the rate constants for force decay after a small stretch are generally much slower than the rate constant for myosin subfragment-1 detachment from actin in solution. It also does not explain why, in the presence of ATP analogue at ionic strength  $\sim 100$  mM, the relationship between the rate constants for force decay after a small stretch and nucleotide analogue concentration is not described by the Michaelis-Menten equation (Anderson and Schoenberg, 1987).

The Schoenberg (1985) model characterized the cross-bridge as single-headed, with all the heads acting independently. Tozeren and Schoenberg (1986) examined the simplest double-headed cross-bridge model, one in which, again, the heads were independent. That model also failed to explain the analogue data. The key additional assumption made by Anderson and Schoenberg (1987), Tozeren (1987), and in the current work

that leads to an explanation of the ATP analogue data is the assumption that the more highly-strained of the two attached cross-bridge heads is unable, after detachment, to reattach in a position of lesser head strain unless the proximally attached head also detaches concurrently. With the independent head model of Tozeren and Schoenberg (1986), the distally attached cross-bridge head is able to quickly detach, swing around the proximally attached head and reattach in a position of lesser strain, quickly relieving the tension it supports. In other words, in the model of Tozeren and Schoenberg (1986), the attached cross-bridge heads act very much as if they were independent. Under conditions where the heads of the cross-bridge bind tightly, the nonindependent-head assumption of Anderson and Schoenberg (1987) leads to a greatly reduced rate constant of force decay after stretch, corresponding to the situation seen experimentally. Furthermore, because the two attached heads no longer act independently, the sigmoidicity in the nucleotide analogue concentration dependence of the rate constant is also explained.

The model of Anderson and Schoenberg (1987) adheres to the central foundation of Schoenberg (1985) that a crossbridge head may basically be characterized solely in terms of the crossbridge strain and the strain-dependent attachment and detachment rate constants. Thus, the model has all the virtues of the Schoenberg (1985) model, and additionally, at the cost of slightly increased complexity, resolves the above mentioned difficulties that the simpler model has in explaining the behavior of cross-bridges in the presence of ATP analogues.

There are two major predictions of the current model that differ from predictions of the 1985 model. The first of these is the prediction that the rate constant for force decay after a small stretch of rabbit psoas fiber should be very dependent upon ionic strength. This prediction follows directly from the current model but not the previous model because head binding in rabbit skeletal muscle varies greatly with ionic strength (Greene et al., 1983) but the detachment rate constants do not (Marston, 1982; Konrad and Goody, 1982). A second prediction of the current model that differs from the previous simpler model is the prediction that at moderate and low ionic strength there should be a sigmoidal relationship between the rate constant of force decay after a small stretch and nucleotide analogue concentration whereas at high ionic strength, there should not be.

The essence of the current model that is responsible for its success is the assumption that once the first head of the cross-bridge binds, there is a change in the locus of actin sites available for binding of the second head. It is reasonable to ask whether there is any plausible model

of the cross-bridge that might give this sort of behavior. One simple, but by no means exclusive, model that might give this kind of behavior is one in which, when the cross-bridge head attaches, it attaches first at  $\sim 90^\circ$  and then rotates to  $\sim 45^\circ$ . It is easy to imagine that whereas the “second” head might initially be able to attach to site  $-1$ ,  $0$ , or  $+1$ , after the “first” head attaches to site  $0$  at  $90^\circ$  and rotates to  $45^\circ$ , the second head might be swept too far away from the  $-1$  actin site to still bind there. Whereas there is some evidence that the crossbridge head undergoes a conformational change upon binding (Trybus and Taylor, 1980, 1982; Coates et al., 1985), there is no evidence that the conformational change that occurs is one that indeed drags the second head away from the proximal actin site.

There is no firm proof of the current model. Its usefulness will depend, as always, upon its ability to continue to explain and predict future experimental results.

## APPENDIX I

This appendix contains the FORTRAN program used to calculate the response of the double-headed cross-bridge model. It calculates the force response to a step stretch of size  $dx$ . The program prompts for  $dx$  with the query “Step size?”. Because of restrictions imposed by the FORTRAN language, not all of the symbols in the program are the same as in the text. For example,  $k_-$  is  $km$ ,  $k_+$  is  $kp$ ,  $k_b$  is  $kb$ , and  $k_f$  is  $kf$ . The forward and back rate constants of each of the steps in Scheme I are referred to as  $rf(m, k)$  and  $rb(m, k)$ , where “ $m$ ” references the actin site and “ $k$ ” the actomyosin state.  $y(k)$  gives the fraction of cross-bridges in each of the six states. The rate constants used for the calculations are found in the subroutine RATES. In the calculations, change of cross-bridge head binding constant was accomplished by changing the value of  $kf$ . This insured that  $k_-$  was always  $\gg k_+$ . The program shown is the program to calculate the case where both heads contribute an equal amount to overall cross-bridge stiffness. Comments explain how to change the program for the situation where the two-headed species have the same stiffness as the one-headed ones. DGEAR, a proprietary subroutine which solves systems of ordinary differential equations using Gear’s method (Gear, 1971), is part of the IMSL library of subroutines.

```

C      EQUILIBRIUM MUSCLE BEHAVIOR
C      This program uses multiple sites and computes the
C      behavior of a double-headed crossbridge. It is assumed
C      that the steps involved are one-step binding of nucleotide
C      and one step detachment of heads (2 for each). The assumption
C      is made that if you know where the first head binds
C      you know where the second head will bind. It is assumed the
C      force origin of the two-headed species is the same as the
C      one-headed but that the stiffness is double.
C
C      Written by Mark Schoenberg, NIH, 301-496-1023, based upon a program
C      originally written by Aydin Tozeren.
C
      IMPLICIT DOUBLE PRECISION(A-H,O-Z)
      DOUBLE PRECISION Y(30),WK (1440)
      DOUBLE PRECISION TIME(400),p(5,6),Q(5,6),Rf(5,6),rb(5,6)
      DOUBLE PRECISION FORCE(20,400),f(400)
      INTEGER IWK(30)
      EXTERNAL FCN,FCNJ,RATES
      common x,s,fk,cn,na,ns,x0,dx,tr

CCCCC  Units: s, nm; fk, dyn/cm, but program divides by 3.8; t, ms;

CCCCC  integer convention, I for x; J for t; M for actin; K for state;
CCCCC  s is separation between actins in nm.
      s = 5.5
CCCCC  na = number of actins (odd); ns = number of attached states;
CCCCC  nx = number of x-divisions used in 'integration
      na = 3
      ns = 5
      nx = 10
      nxpl = nx + 1
      fk = 1.
      fk = fk/3.8
      write (6,102)
102    format (" Nucleotide conc.?" )
      read(5,*)cn
      write (6,104)

```

```

104  format (" Step size?")
      read(5,*)dx
      write(6,106)
106  format(" dt (ms)?")
      read(5,*)dt
CCCCC Below are parameters used by DGEAR
      MM= 50
      TOL=0.0001
      H=0.00001
      METH=1
      MITER=1
      N= na*ns + 1
CCCCC      Start of main x loop (outer loop variable)
      x = -s/2 - s/nx
      do 90 i = 1,nxp1
      x = x + s/nx
      t = 0.
      CALL RATES (rf,rb,t)
      do 10 m = 1,na
      do 10 k = 1, (ns+1)
10    q(m,k) = rf(m,k)/rb(m,k)
      do 20 m = 1,na
      o = m - (na+1)/2
CCCCC      Factor of two below is due to stiffness = 2*fk for 2 heads.
      p(m,1) = dexp(-2.*fk*0.5*(x+o*s)**2)
      p(m,2) = p(m,1)*q(m,1)
      p(m,3) = p(m,1)*q(m,1)*q(m,2)
      p(m,4) = p(m,1)*q(m,1)*q(m,3)
20    p(m,5) = p(m,1)*q(m,1)*q(m,2)*q(m,4)
      pu = p((na+1)/2,1)*q((na+1)/2,1)*q((na+1)/2,2)
      + *q((na+1)/2,4)*q((na+1)/2,6)
      pt = pu
      do 30 m = 1,na
      do 30 k = 1,ns
30    pt = pt + p(m,k)
      y(ns*na+1) = pu/pt

CCCCC      Start of main time loop (inner variable)
70    INDEX=1
      CALL FORFUN(n,y,force(i,1))
      do 80 j =2,mm
      xend=t + dt
      CALL DGEAR(N,FCN,FCNJ,t,H,Y,XEND,TOL,METH,MITER,INDEX,IWK,WK,IER)
      Time(j) = t
      CALL FORFUN(n,y,force(i,j))
80    continue
CCCCC      End of time loop
90    continue
CCCCC      End of x loop
      Time(1)=0.
      WRITE(1,112)mm
112    FORMAT(I7)
      do 210 j = 1,mm
      f(j) = 0.
      do 200 i = 1,nxp1
      f(j) = f(j) + force(i,j)/nx
      f(j) = f(j) - 0.5*(force(1,j) + force(nxp1,j))/nx
      WRITE(1,114) Time(j),f(j)
200    continue
114    FORMAT(2F15.7)
CCCCC      Next four lines for calculating the half time of force decay, tau.
      if (j .lt. 2) goto 210
      if ((F(j).le.(F(1)/2.)).AND.(F(j-1).gt.(F(1)/2.)))
1    tau=time(j)+(time(j-1)-time(j))*((F(1)/2.)-F(j))/
1    (F(j-1)-F(j))

```

```

210    CONTINUE
CCCCCC Below calculates the rate constant of force decay.
      if (tau .gt. 0.) rate = 0.69/tau
      write(6,115)cn,F(1),tau,rate
115    format(4f12.3)
      STOP
      END

      SUBROUTINE FCN(N,T,Y,YPRIME)
      IMPLICIT DOUBLE PRECISION(A-H,O-Z)
      DOUBLE PRECISION Y(N),YPRIME(N)
      DOUBLE PRECISION rf(5,6),rb(5,6)
      common x,s,fk,cn,na,ns,x0,dx,tr

      CALL RATES(rf,rb,t)
      yprime(ns*na+1) = 0.
CCCCCC The above gets treated differentially since detached can go to
CCCCCC any actin.
      do 5 m = 1,na
      mi = ns*(m-1)
      yprime(mi+1) = -rf(m,1)*y(mi+1)+rb(m,1)*y(mi+2)
      yprime(mi+2) = -(rb(m,1)+rf(m,2)+rf(m,3))*y(mi+2)
      + rf(m,1)*y(mi+1)+rb(m,3)*y(mi+4)+rb(m,2)*y(mi+3)
      yprime(mi+3) = -(rb(m,2)+rf(m,4))*y(mi+3)
      + rf(m,2)*y(mi+2)+rb(m,4)*y(mi+5)
      yprime(mi+4) = -(rb(m,3)+rf(m,5))*y(mi+4)
      + rf(m,3)*y(mi+2)+rb(m,5)*y(mi+5)
      yprime(mi+5) = -(rb(m,5)+rb(m,4)+rf(m,6))*y(mi+5)
      + rf(m,5)*y(mi+4)+rf(m,4)*y(mi+3)+rb(m,6)*y(ns*na+1)
      yprime(ns*na+1)=yprime(ns*na+1)-rb(m,6)*y(ns*na+1)
      + rf(m,6)*y(mi+5)
5      continue
      RETURN
      END

      SUBROUTINE FCNJ(N,T,Y,PD)
      IMPLICIT DOUBLE PRECISION(A-H,O-Z)
      INTEGER N
      DOUBLE PRECISION Y(N),PD(N,N),T
      DOUBLE PRECISION rf(5,6),rb(5,6)
      common x,s,fk,cn,na,ns,x0,dx,tr
      CALL RATES(rf,rb,t)
      do 5 i = 1,N
      do 5 ii = 1,N
5      pd(i,ii) = 0.
      do 10 m = 1,na
      mi = ns*(m-1)
      pd(mi+1,mi+1) = -rf(m,1)
      pd(mi+1,mi+2) = rb(m,1)
      pd(mi+1,mi+3) = 0.
      pd(mi+1,mi+4) = 0.
      pd(mi+1,mi+5) = 0.
      pd(mi+2,mi+1) = rf(m,1)
      pd(mi+2,mi+2) = -(rb(m,1)+rf(m,2)+rf(m,3))
      pd(mi+2,mi+3) = rb(m,2)
      pd(mi+2,mi+4) = rb(m,3)
      pd(mi+2,mi+5) = 0.
      pd(mi+3,mi+1) = 0.
      pd(mi+3,mi+2) = rf(m,2)
      pd(mi+3,mi+3) = -(rb(m,2)+rf(m,4))
      pd(mi+3,mi+4) = 0.
      pd(mi+3,mi+5) = rb(m,4)
      pd(mi+4,mi+1) = 0.

```

```

pd(mi+4,mi+2)=rf(m,3)
pd(mi+4,mi+3)=0.
pd(mi+4,mi+4)= -(rb(m,3)+rf(m,5))
pd(mi+4,mi+5)=rb(m,5)
pd(mi+5,mi+1)=0.
pd(mi+5,mi+2)=0.
pd(mi+5,mi+3)=rf(m,4)
pd(mi+5,mi+4)=rf(m,5)
pd(mi+5,mi+5)= -(rb(m,5)+rb(m,4)+rf(m,6))
pd(mi+5,ns*na+1)=rb(m,6)
pd(ns*na+1,mi+5)=rf(m,6)
pd(ns*na+1,ns*na+1)=pd(ns*na+1,ns*na+1) - rb(m,6)
CCCCCC Note that above is a little different due to special nature
CCCCCC of off state.
10 continue
RETURN
END

```

```

SUBROUTINE RATES(rf,rb,t)
IMPLICIT DOUBLE PRECISION(A-H,O-Z)
DOUBLE PRECISION rf(5,6),rb(5,6),t
double precision kp,km,kf,kb
common x,s,fk,cn,na,ns,x0,dx,tr
xx=0.
if (t .gt. 0.) xx = dx
do 5 m = 1,na
o = m - (na+1)/2
kp = 333.
km = 1000.
kb = 1.
kf = 0.01
rf(m,1) = 2*kp*cn
rb(m,1) = km
rf(m,2) = kp*cn
rb(m,2) = 2.*km
rf(m,3) = kb
rb(m,3) = kf*dexp(-0.5*fk*(x+o*s+xx)**2)
rf(m,4) = 2.*kb
rb(m,4) = kf*dexp(-0.5*fk*(x+o*s+xx)**2)
rf(m,5) = kp*cn
rb(m,5) = km
rf(m,6) = kb
5 rb(m,6) = 2*kf*dexp(-0.5*fk*(x+o*s+xx)**2)
RETURN
END

```

```

SUBROUTINE FORFUN(n,y,f)
IMPLICIT DOUBLE PRECISION(A-H,O-Z)
integer n
double precision y(n),f
common x,s,fk,cn,na,ns,x0,dx,tr

```

```

CCCCCC A subroutine to calculate force when states have y(n) occupancy.
f = 0.
do 6 m = 1,na
o = m - (na+1)/2
do 5 k = 1,ns
fkcor = 1.
CCCCCC Next line goes out if second head adds nothing to overall stiffness
if (k .le. 3) fkcor = 2.
5 f = f + y(ns*(m-1)+k)*fk*(x+o*s+dx)*fkcor
6 continue
RETURN
END

```

I am greatly indebted to Dr. Terrell L. Hill, both for laying down the theoretical foundation for the ideas expressed herein and for patiently explaining things to me when I first became involved in this area of study.

Received for publication 14 November 1990 and in final form 19 April 1991.

## REFERENCES

- Anderson, M., and M. Schoenberg. 1987. Possible cooperativity in crossbridge detachment in muscle fibers having magnesium pyrophosphate at the active site. *Biophys. J.* 52:1077-1082.
- Clarke, M. L., and R. T. Tregear. 1982. Tension maintenance and crossbridge detachment. *FEBS (Fed. Eur. Biochem. Soc.) Lett.* 143:217-219.
- Coates, J. H., A. H. Criddle, and M. A. Geeves. 1985. Pressure-relaxation studies of pyrene-labelled actin and myosin subfragment 1 from rabbit skeletal muscle: evidence for two states of acto-subfragment 1. *Biochem. J.* 232:351-356.
- Gear, C. W. 1971. Numerical initial value problems in ordinary differential equations. Prentice-Hall, Inc. Englewood Cliffs, New Jersey.
- Greene, L. E., J. R. Sellers, E. Eisenberg, and R. S. Adelstein. 1983. Binding of gizzard smooth muscle myosin subfragment 1 to actin in the presence and absence of adenosine 5'-triphosphate. *Biochemistry.* 22:530-535.
- Hill, T. L. 1974. Theoretical formalism for the sliding filament model of contraction of striated muscle. Part I. *Prog. Biophys. Molec. Biol.* 28:267-340.
- Hill, T. L. 1975. Theoretical formalism for the sliding filament model of contraction of striated muscle. Part II. *Prog. Biophys. Molec. Biol.* 29:105-159.
- Konrad, M., and R. S. Goody. 1982. Kinetic and thermodynamic properties of the ternary complex between F-actin, myosin subfragment 1 and adenosine 5'-[ $\beta,\gamma$ -imido]triphosphate. *Eur. J. Biochem.* 128:547-555.
- Marston, S. B. 1982. The rates of formation and dissociation of actin-myosin complexes. *Biochem. J.* 203:453-460.
- Schoenberg, M. 1985. Equilibrium muscle crossbridge behavior: theoretical considerations. *Biophys. J.* 48:467-475.
- Schoenberg, M. 1988a. The kinetics of weakly- and strongly-binding crossbridges: implications for contraction and relaxation. *Adv. Exp. Med. Biol.* 226:189-202.
- Schoenberg, M. 1988b. Characterization of the myosin adenosine triphosphate (M·ATP) crossbridge in rabbit and frog skeletal muscle fibers. *Biophys. J.* 54:135-148.
- Schoenberg, M. 1989. Effect of adenosine triphosphate analogues on skeletal muscle fibers in rigor. *Biophys. J.* 56:33-41.
- Schoenberg, M. and E. Eisenberg. 1985. Muscle cross-bridge kinetics in rigor and in the presence of ATP analogues. *Biophys. J.* 48:863-871.
- Tozeren, A. 1987. The influence of doubly attached crossbridges on the mechanical behavior of skeletal muscle fibers under equilibrium conditions. *Biophys. J.* 52:901-906.
- Tozeren, A., and M. Schoenberg. 1986. The effect of cross-bridge clustering and head-head competition on the mechanical response of muscle fibers under equilibrium conditions. *Biophys. J.* 50:875-884.
- Trybus, K. M., and E. W. Taylor. 1980. Kinetic studies of the cooperative binding of subfragment 1 to regulated actin. *Proc. Natl. Acad. Sci. USA.* 77:7209-7213.
- Trybus, K. M., and E. W. Taylor. 1982. Transient kinetics of adenosine 5'-diphosphate and adenosine 5'-( $\beta,\gamma$ -imidotriphosphate) binding to subfragment 1 and actosubfragment 1. *Biochemistry.* 21:1284-1291.